Verzeichnisse und Dateien per VBA auslesen und verarbeiten, Teil 1
Versionen: Access 2003, 2002/XP, 2000 und 97

In Ihren Datenbanken müssen Sie immer wieder eine Auswahl der Dateien eines Verzeichnisses anzeigen oder Datei-Informationen in Tabellen zu speichern. Unser zweiteiliger Tipp der Woche liefert Ihnen die dazu notwendigen Grundlagen und Funktionen.

Basis einer Funktion für das Auslesen von Verzeichnissen ist die Funktion "Dir()": Sie wird zunächst mit dem Pfad des auszulesenden Verzeichnisses und einer Dateimaske, die die in Frage kommenden Dateien spezifiziert, aufgerufen:

strFName = Dir("Z:\Test\*.mdb")

Das Ergebnis ist entweder eine leere Zeichenkette (""), wenn keine passende Datei gefunden wurde oder der Dateiname der ersten, zur Dateimaske passenden Datei:

? Dir("Z:\Test\*.mdb") 
Adressen.mdb

Nach diesem ersten, initialisierenden Aufruf wird "Dir()" in einer Schleife solange aufgerufen, bis die Rückmeldung eine leere Zeichenkette ("") ist. Das Ergebnis der einzelnen Aufrufe in der Schleife ist die jeweils nächste passende Datei. Mehr dazu gleich. In der Dateimaske können Sie zwei so genannte "Wildcards" verwenden und damit das Ergebnis auf bestimmte Dateigruppen einschränken. Das Zeichen "*" (Sternchen) steht dabei als Platzhalter für eine beliebige Zeichenfolge bzw. eine beliebige Anzahl von Zeichen. Das Zeichen "?" (Fragezeichen) steht als Platzhalter für ein einzelnes Zeichen. Die Dateimaske "*.mdb" bedeutet also "Egal, welcher Dateiname, die Erweiterung muss jedoch 'mdb' lauten" und liefert somit alle Access-Datenbanken. Analog dazu liefert eine Dateimaske "*.xls" alle Excel-Arbeitsmappen oder "*.doc" alle Word-Dokumente. Eine Dateimaske wie "Adr*.mdb" definiert alle Dateien, die mit den Zeichen "Adr" beginnen, was dann folgt, ist wieder egal, die Dateierweiterung muss jedoch 'mdb' lauten. Das Ergebnis ist dann beispielsweise "Adressen.mdb", "Adressverwaltung.mdb" oder "Adressen Kunden.mdb". Mit dem Wildcard "?" können Sie eine Dateigruppe spezifizieren, die vom Dateinamen oder der Erweiterung her gleich aufgebaut ist, innerhalb des Dateinamens oder der Erweiterung aber spezielle Merkmale aufweist, die von Interesse sind.

So liefert beispielsweise die Dateimaske "Budget ??-2005.xls" alle Excel-Arbeitsmappen monatlich erstellter Budgetplanungen wie "Budget 01-2005.xls", "Budget 02-2005.xls", "Budget 03-2005.xls" usw., ignoriert aber Dateien wie "Budget Einkauf 2005.xls" oder "Budget Quartal II 2005.xls". Analog dazu definieren Sie zum Beispiel mit "Produktinfos*.do?" eine Dateigruppe, bei der alle Dateinamen mit "Produktinfos" beginnen und die ersten beiden Zeichen der Erweiterung "do" aufweisen. Das Ergebnis sind alle Word-Dokumente (.doc) und Word-Vorlagen (.dot) wie "Produktinfos DigiCam D850XL.doc" oder "Produktinfos Blanco.dot".

Wie bereits erwähnt, erfolgt der Einsatz der "Dir()"-Funktion innerhalb einer Schleife wie in folgendem Beispiel:

Sub DirTest()
  Dim strPath As String
  Dim strFName As String

  strPath = "Z:\Test\"
  strFName = Dir(strPath & "*.mdb")
  While strFName <> ""
    Debug.Print strFName, _
                FileLen(strPath & strFName), _
                FileDateTime(strPath & strFName)
    strFName = Dir()
  Wend
  Beep

End Sub

Über die Variable "strPath" legen wir hier zunächst das Verzeichnis fest, aus dem Dateiinfos ausgelesen werden sollen. Der erste Aufruf 'strFName = Dir(strPath & "*.mdb")' initialisiert "Dir()" und liefert als Ergebnis entweder eine leere Zeichenkette ("") oder den Namen der ersten gefundenen Datei. Die folgende While/Wend-Schleife wird also nur ausgeführt, wenn mindestens eine passende Datei gefunden wurde. Deren Namen sowie Infos zur Dateigröße und zu Datum/Zeit der letzten Änderung geben wir dann im Direktbereich aus. Der folgende Aufruf 'strFName = Dir()' liefert dann die nächste passende Datei - oder eine leere Zeichenkette, wenn keine weiteren passenden Dateien mehr vorhanden sind und terminiert somit die While-/Wend-Schleife. Ein abschließendes 'Beep' informiert akustisch darüber, dass die Routine ihre Aufgabe erledigt hat.

Mit diesen Grundlagen gerüstet können Sie nun beispielsweise Dateiinfos aus einem bestimmten Verzeichnis in eine Access-Tabelle einlesen:

Sub DirTest_2()
  Dim strPath As String
  Dim strFName As String
  Dim db As DAO.Database, rs As DAO.Recordset

  Set db = CurrentDb()
  Set rs = db.OpenRecordset("Dateiinfos", dbOpenDynaset)
  strPath = "Z:\Test\"
  strFName = Dir(strPath & "*.mdb")
  While strFName <> ""
    rs.AddNew
    rs("Dateiname") = strPath & strFName
    rs("Dateigröße") = FileLen(strPath & strFName)

    rs("DateiDatumZeit") = FileDateTime( _
    strPath & strFName)

    rs.Update
    strFName = Dir()
  Wend
  rs.Close
  Set rs = Nothing
  Set db = Nothing
  Beep

End Sub

Dieses Beispiel initialisiert zunächst einen Recordset "rs" für den Zugriff auf die Tabelle "Dateiinfos". Diese Tabelle umfasst die folgenden Felder:

Dateiname: Text, Länge = 250
Dateigröße: Zahl/Long Integer
DateiDatumZeit: Datum/Zeit

Anschließend werden in einer Schleife alle Dateien mit der Erweiterung 'mdb' ausgelesen. Mit "rs.AddNew" fügen wir der Tabelle einen neuen Datensatz hinzu, dessen Felder dann mit den Dateiinfos 'strPath & strFName' (Pfad und Dateiname), 'FileLen(strPath & strFName)' (Dateigröße) und 'FileDateTime(strPath & strFName)' (Datum/Zeit letzte Änderung) gefüllt werden. 'rs.Update' schreibt den Datensatz danach in die Tabelle. Sind alle Dateien ausgelesen, wird der Recordset wieder geschlossen - fertig! Die Dateiinfos befinden sich nun in der Tabelle.

Wie Sie nun die Dateiinfos aus dieser Tabelle beispielsweise zur Auswahl in einer Liste anzeigen oder eine Auswahl ohne Zwischenspeicherung in einer Tabelle realisieren, ist Thema des zweiten Teils in der nächsten Ausgabe von Access Weekly.

Im ersten Teil dieses Tipps der Woche hatten wir erläutert, wie alle Dateien (*.*) oder bestimmte Dateigruppen (*.mdb) aus einem Verzeichnis ausgelesen (Funktion "Dir()") und in einer Tabelle gespeichert werden. Dabei wurden neben Pfad/Dateinamen auch Informationen zur Dateigröße (Funktion "FileLen()") und zum letzten Änderungsdatum (Funktion "FileDateTime()") erfasst. Die so gewonnenen Daten lassen sich nun beispielsweise für die Anzeige in einer Dateiauswahl heranziehen. Dazu legen Sie ein einem Formular ein Listenfeld (Name= "lstDatAusw") an und stellen dessen Eigenschaft "Datensatzherkunft" auf die folgende SQL-Anweisung ein:

select Dateiname from DateiInfos order by Dateiname;

Diese Dateiauswahl kann nun zum Beispiel dazu genutzt werden, den Anwender eine Excel-Arbeitsmappe für den Import auswählen zu lassen. Dazu legen Sie unter der Liste eine Schaltfläche Importieren (Name= btnImport) an und geben für deren Ereignisprozedur "Beim Klicken" die folgenden Anweisungen ein:

Private Sub btnImport_Click()
  Dim strFName As String

  strFName = Me.lstDatAusw & ""
  If strFName = "" Then
    Beep
    Me.lstDatAusw.SetFocus
    Exit Sub
  End If

  DoCmd.TransferSpreadsheet _
        acImport, _
        acSpreadsheetTypeExcel97, _
        "ExcelImport", _
        strFName

End Sub

Diese Anweisungen prüfen zunächst, ob in der Liste ein Eintrag ausgewählt ist. Wenn nicht, wird ein akustischer Hinweis ausgegeben, die Liste aktiviert und die Routine verlassen. Andernfalls erfolgt der Import (acImport) der ausgewählten Arbeitsmappe (strFName, acSpreadsheetTypeExcel97) per "DoCmd.TransferSpreadsheet" in die Tabelle "ExcelImport".

Möchten Sie in der Dateiauswahl außer Pfad/Dateinamen noch Infos zur Größe und dem Änderungsdatum anzeigen, stellen Sie die Liste einfach auf mehrspaltig ein ("Spaltenanzahl" = 3), setzen die gewünschten Breiten der Spalten (zum Beispiel "Spaltenbreiten" = "5cm;2cm;2cm") und ändern "Datensatzherkunft" auf die folgende SQL-Anweisung:

select * from DateiInfos order by Dateiname;

Um eine solche Dateiauswahl zu realisieren, müssen die Dateiinfos zunächst in eine Tabelle eingelesen werden. Um diesen Zwischenschritt zu vermeiden, können Sie eine so genannte "CallBack-Funktion" einsetzen und darüber die Dateiauswahl, sprich die Liste "lstDatAusw", füllen lassen. Die Callback-Funktion besteht aus zwei Teilen, einer Funktion "DatAuswAufbauen", die Pfad/Dateinamen zunächst in ein Array einliest und der eigentlichen CallBack-Funktion "DatAuswSetzen", die der Liste über deren Eigenschaft "Herkunftstyp" zugeordnet wird und die die Inhalte in die Liste überträgt. Geben Sie zunächst im allgemeinen Teil des Formulars die folgenden Deklarationen und Funktionen ein:

Dim arrDatAusw() As String
Dim intDatAusw As Integer

Function DatAuswAufbauen()
  Dim strPath As String, strFName As String
  Dim R As Variant

  DatAuswAufbauen = False
  Erase arrDatAusw()
  intDatAusw = 0

  strPath = "Z:\Test\"
  strFName = Dir(strPath & "*.xls")
  While strFName <> ""
    intDatAusw = intDatAusw + 1
    ReDim Preserve arrDatAusw(intDatAusw)
    arrDatAusw(intDatAusw) = strPath & strFName
    strFName = Dir()
  Wend

  DatAuswAufbauen = True

End Function

Function DatAuswSetzen(Feld As Control, _
                       ID As Variant, _
                       Zeile As Variant, _
                       Spalte As Variant, _
                       Code As Variant) As Variant

  On Error Resume Next
  Select Case Code

    Case acLBInitialize ' Initialisieren.
      DatAuswSetzen = DatAuswAufbauen()

    Case acLBOpen ' Öffne mit eindeutiger ID
      DatAuswSetzen = Timer

    Case acLBGetRowCount 'Wieviel Zeilen hat die Liste?
      DatAuswSetzen = intDatAusw

    Case acLBGetColumnCount 'Wieviel Spalten?
      DatAuswSetzen = 1

    Case acLBGetColumnWidth ' Spaltenbreite, hier Standard
      DatAuswSetzen = -1

    Case acLBGetValue 'Daten für die Liste liefern
      DatAuswSetzen = arrDatAusw(Zeile + 1)

  End Select

End Function

Stellen Sie dann die Eigenschaft "Herkunftstyp" der Liste auf "DatAuswSetzen" ein und löschen Sie eventuell vorhandene Inhalte in "Datensatzherkunft". Wenn Sie das Formular nun einmal anzeigen lassen, sehen Sie eine Dateiauswahl, die ohne Umwege über eine Tabelle die Dateien eines Verzeichnisses anzeigt.

Und so funktioniert das Ganze: Wenn das Formular angezeigt wird, ruft Access für das Füllen der Liste zunächst "DatAuswSetzen" mit dem Parameter "acLBInitialize" auf, so dass "DatAuswAufbauen" aufgerufen wird und die Dateien einliest. Der zweite und dritte Aufruf mit den Parametern "acLBGetRowCount" und "acLBGetColumnCount" ermittelt dann die Anzahl Einträge für die Liste, die der zuvor in "intDatAusw" festgehaltenen Anzahl Dateien entspricht und die Anzahl Spalten, die hier "1" ist. Weitere Aufrufe mit "acLBGetValue" liefern dann die Inhalte für die einzelnen Einträge der Liste aus dem Array "arrDatAusw".

